<commit_before>{spawn} = require 'child_process'
binding = process.atomBinding 'crash_reporter'

class CrashReporter
  start: (options={}) ->
    {productName, companyName, submitUrl, autoSubmit, ignoreSystemCrashHandler, extra} = options

    productName ?= 'Atom-Shell'
    companyName ?= 'GitHub, Inc'
    submitUrl ?= 'http://172.16.58.3:1127/post'
    autoSubmit ?= true
    ignoreSystemCrashHandler ?= false
    extra ?= {}

    extra._productName ?= productName
    extra._companyName ?= companyName
    extra._version ?=
      if process.__atom_type is 'browser'
        require('app').getVersion()
      else
        require('remote').require('app').getVersion()

    start = -> binding.start productName, companyName, submitUrl, autoSubmit, ignoreSystemCrashHandler, extra

    if process.platform is 'darwin'
      start()
    else
      args = [
        "--reporter-url=#{submitUrl}"
        "--application-name=#{productName}"
        "--v=1"
      ]
      env = ATOM_SHELL_INTERNAL_CRASH_SERVICE: 1

      spawn process.execPath, args, {env, detached: true}
      start()

module.exports = new CrashReporter<commit_msg>:lipstick: Wrong IP for default crash reporter server.<commit_after>{spawn} = require 'child_process'
binding = process.atomBinding 'crash_reporter'

class CrashReporter
  start: (options={}) ->
    {productName, companyName, submitUrl, autoSubmit, ignoreSystemCrashHandler, extra} = options

    productName ?= 'Atom-Shell'
    companyName ?= 'GitHub, Inc'
    submitUrl ?= 'http://192.168.3.11:1127/post'
    autoSubmit ?= true
    ignoreSystemCrashHandler ?= false
    extra ?= {}

    extra._productName ?= productName
    extra._companyName ?= companyName
    extra._version ?=
      if process.__atom_type is 'browser'
        require('app').getVersion()
      else
        require('remote').require('app').getVersion()

    start = -> binding.start productName, companyName, submitUrl, autoSubmit, ignoreSystemCrashHandler, extra

    if process.platform is 'darwin'
      start()
    else
      args = [
        "--reporter-url=#{submitUrl}"
        "--application-name=#{productName}"
        "--v=1"
      ]
      env = ATOM_SHELL_INTERNAL_CRASH_SERVICE: 1

      spawn process.execPath, args, {env, detached: true}
      start()

module.exports = new CrashReporter
{'fraction_non_alphanumeric': 0.12435008665511266, 'fraction_numerical': 0.012564991334488735, 'mean_word_length': 3.2058287795992713, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 7, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 2, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '4315394', 'n_tokens_mistral': 723, 'n_tokens_neox': 695, 'n_words': 150}